
*
*    R0A= 1/A0A                   R0B= 1/A0B                 R0C= 1/A0C
*   _/\  /\_                 _/\  /\__                    _/\  /\_
*   |  \/   |    R3A= 1/A0A  |  \/    |       R3B= 1/A0B |   \/   |  R3C= 1/A0C
*   |     __|_/\  /\  /\_____|_____   |_/\  /\  /\_______|_____   |_/\  /\  /\_____________
*   |     |     \/  \/          LPA|  |   \/  \/            LPB|  |   \/  \/            LPC|
*  _|_    |    R2A= 1/A1A          |  |    R2B= 1/A1B          |  |    R2C= 1/A1C          |
* /VIN\   |_/\  /\  /\_____        |  |_/\  /\  /\_____        |  |_/\  /\  /\_____        |
* \___/   |   \/  \/       |       |  |   \/  \/       |       |  |   \/  \/       |       |
*   |     |   R1A= 1       |       |  |   R1 = 1       |       |  |   R1C= 1       |       |
*  _|_    |_/\  /\_        |BPA    |  |_/\  /\_        | BPB   |  |_/\  /\_        | BPC   |
*  ///    |   \/   |       |       |  |   \/   |       |       |  |   \/   |       |       |
*         | |\  HPA|XONES1A|XONES2A|  | |\  HPB|XONES1B|XONES2B|  | |\  HPC|XONES1C|XONES2C|
*         |_|-\    | ____  | ____  |  |_|-\    | ____  | ____  |  |_|-\    | ____  | ____  |
*       SUMA|  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|
*           |  /     |___|   |___| SUMB |  /     |___|   |___|  SUMC|  /     |___|   |___|
*         __|+/ OPAA  _|_     _|_     __|+/ OPAB  _|_     _|_     __|+/ OPAC  _|_     _|_
*        |  |/        \ /     \ /    |  |/        \ /     \ /    |  |/        \ /     \ /
*       _|_            V       V    _|_            V       V    _|_            V       V
*       ///              VBIAS      ///              VBIAS      ///              VBIAS
*      Butterworth terms
*      (s^2 + 0.5176s + 1)(s^2 + 1.4142s + 1)(s^2 + 1.9319s + 1)
.OPTIONS  GMIN=1e-18    METHOD=gear   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN    0      PULSE( -.5 .5 1u 1u  1u  10m 20m ) AC = 1
R0A       VIN    SUMA   1    
R1A       SUMA   HPA    1
R2A       SUMA   BPA    1.9319
R3A       SUMA   LPA    1
XOPA1A    SUMA          HPA    OPA
XONES1A   HPA    BPA    VBIAS  ONE_S
XONES2A   BPA    LPA    VBIAS  ONE_S
R0B       LPA    SUMB   1    
R1B       SUMB   HPB    1
R2B       SUMB   BPB    .707
R3B       SUMB   LPB    1
XOPA1B    SUMB          HPB    OPA
XONES1B   HPB    BPB    VBIAS  ONE_S
XONES2B   BPB    LPB    VBIAS  ONE_S
R0C       LPB    SUMC   1    
R1C       SUMC   HPC    1
R2C       SUMC   BPC    .5176
R3C       SUMC   LPC    1
XOPA1C    SUMC          HPC    OPA
XONES1C   HPC    BPC    VBIAS  ONE_S
XONES2C   BPC    LPC    VBIAS  ONE_S

E_INVERT  OUT0   0      LPC    0   -1
VBIAS     VBIAS  0      DC     1

*#0 =========A_Bessel_is_Best_for_Low_Phase_Distortion====
.control
set          pensize =    2
*#1 =========SET_VCF_VBIAS_TO_1=====================
set          outfile0 =             "Butterworth_6P_VCF_1.txt"
tran         1m     40m      0  10m
run
plot         vin out0  title Vbias_1

let          saveData = 1
if           (saveData>0)
echo         "VpwlA  OUTA 0 PWL("    >$outfile0
let          NoOfTime   =   length(time) 
echo        "Number of points is $&NoOfTime "
let          n = 0
repeat       $&NoOfTime
let          timestep = time[n] - time[n-1]
let          timme =  time[n]
let          vout  =  out0[n]
if           (timestep > 1u)
echo         "+ $&timme $&vout "   >> $outfile0
endif
let          n = n+1
end
echo        "+ )"                   >>$outfile0
endif
ac           dec    50     1   10000
run
plot         db(out0) ylimit -20 0   title Vbias_1



*#2 ============SET_VCF_VBIAS_TO_100m================
alter        vbias dc = .1
set          outfile1 =             "Butterworth_6P_VCF_100m.txt"
tran         1m     40m      0  10m
run
plot         vin out0  title Vbias_100m

let          saveData = 1
if           (saveData>0)
echo         "VpwlB  OUTB 0 PWL("    >$outfile1
let          NoOfTime   =   length(time) 
echo        "Number of points is $&NoOfTime "
let          n = 1
repeat       $&NoOfTime
let          timestep = time[n] - time[n-1]
let          timme =  time[n]
let          vout  =  out0[n]
if           (timestep > 1u)
echo         "+ $&timme $&vout "   >> $outfile1
endif
let          n = n+1
end
echo        "+ )"                   >>$outfile1
endif
ac           dec    50     1   10000
run
plot         db(out0) ylimit -20 0   title Vbias_100m

*#3 ============SET_VCF_VBIAS_TO_30m================
alter        vbias dc = .03
set          outfile2 =             "Butterworth_6P_VCF_30m.txt"
tran         1m     40m      0  10m
run
plot         vin out0  title Vbias_30m

let          saveData = 1
if           (saveData>0)
echo         "VpwlC  OUTC 0 PWL("    >$outfile2
let          NoOfTime   =   length(time) 
echo        "Number of points is $&NoOfTime "
let          n = 0
repeat       $&NoOfTime
let          timestep = time[n] - time[n-1]
let          timme =  time[n]
let          vout  =  out0[n]
if           (timestep > 1u)
echo         "+ $&timme $&vout "   >> $outfile2
endif
let          n = n+1
end
echo        "+ )"                   >>$outfile2
endif
ac           dec    50     1   10000
run
plot         db(out0) ylimit -20 0   title Vbias_30m

.endc



*==================OP_AMP=====================================
.SUBCKT      OPA    IN    OUT
EGN1         OUT    0     IN    0    -10k
.ends
*                    ___     .SUBCKT  OPA  IN  OUT
*                ___|OUT|
*     ___      _|_  |___|     ___   |\
*    |IN |___ /   \          |IN |__|-\    ___
*    |___|   | EGN1|  ---->  |___|  |  \__|OUT|
*           _|-10K |  ---->         |  /  |___|
*          |  \___/               __|+/
*          |____|               _|_ |/ OPA
*              _|_              ///
*              ///
*
*==================GM_C=====================================
.SUBCKT      ONE_S  IN    OUT   VB
BGM          1      0     i = -10*v(IN)*v(VB)
C1           1      0     1m
EBUF         OUT    0     1      0      1
.ends
*                                  ___
*   i = -10*v(IN)*v(VBIAS)   _____|OUT|     .SUBCKT   ONE_S  IN   OUT  VBIAS
*  ___       __________     _|_   |___|     ___   ______    ___
* |VB |_   _|_      |  |__ /   \           |IN |_|      |__|OUT|
* |___| |_/ _ \     |1    | EBUF|   ---->  |___| | 1/s  |  |___|
*  ___    \/ \/    _|_   _|  +1 |   ---->        |______|
* |IN |___/\_/\    ___  |  \___/                    |
* |___|   \___/   C1|   |____|                     _|_
*          _|_     _|_      _|_                   |VB |
*     BGM  ///     \\\      ///                   |___|
*

.end

=====================END_OF_SPICE=======================

A Butterworth is designed for good pass band response
as opposed to the Low Phase Distortion response of
a Bessel

*#1 =========SET_VCF_VBIAS_TO_1=====================
set          outfile0 =             "Bessel_6P_VCF_1.txt"
VBIAS        VBIAS  0      DC     1

The fundamental is a square wave at 25Hz. 
With VBIAS set to 1V, the filter's bandwidth 
is a little above 1kHz. Thats means the 40th
harmonic of the 25Hz is the 3dB point. Up
to the 3dB point, the there is very little
attentuation of the harmonics.



But the output waveform has picked up a definite
ringing compared the same conditions as for
a bessel response.

*#2 =========SET_VCF_VBIAS_TO_100m================
alter        vbias dc = .1
set          outfile1 =             "Bessel_6P_VCF_100m.txt"

When VBIAS is set now to 100m, the bandwidth
of the 6pole filter is now ten times smaller.



There is now about a 4 msec delay between the
input and output waveforms. The 3dB point
is now about at the 4th harmonic of 25Hz.
The output waveform has developed a larger
lower frequency ringing/

*#3 =========SET_VCF_VBIAS_TO_30m================
alter        vbias dc = .03
set          outfile2 =             "Bessel_6P_VCF_30m.txt"
Setting VBIAS down to 30m means all the 
odd harmonics are now attentuated.



The input to output delay has now increased
to about 14msec and the output waveform is
now close to being a pure 25Hz sinewave. 


